

DISPLAY

`DISPLAY [(rep)] [options] {[/ ...] [output-format] output-element } ...`

Dieses Kapitel behandelt folgende Themen:

- Funktion
- Syntax-Beschreibung
- Standardwerte
- Beispiele

Eine Erläuterung der in dem Syntax-Diagramm verwendeten Symbole entnehmen Sie dem Abschnitt *Syntax-Symbole*.

Verwandte Statements: AT END OF PAGE | AT TOP OF PAGE | CLOSE PRINTER | DEFINE PRINTER | EJECT | FORMAT | NEWPAGE | PRINT | SKIP | SUSPEND IDENTICAL SUPPRESS | WRITE | WRITE TITLE | WRITE TRAILER

Gehört zur Funktionsgruppe: *Erstellen von Ausgabe-Reports*

Funktion

Mit dem DISPLAY-Statement werden die Felder angegeben, deren Werte ausgegeben werden sollen. Die Ausgabe erfolgt in Spaltenform, mit einer Spalte pro Feld und einer Spaltenüberschrift.

Anmerkung:

Die Statements WRITE und PRINT können benutzt werden, um Ausgaben in Freiformat (keine Spalten) zu erzeugen.

Siehe auch die folgenden Themen im *Leitfaden zur Programmierung*:

- *Steuerung der Ausgabe von Daten*
- *Statements DISPLAY und WRITE*
- *Index-Notation (n:n) für multiple Felder und Periodengruppen*
- *Spaltenüberschriften*
- *Layout einer Ausgabeseite*

Syntax-Beschreibung

<i>(rep)</i>	<p>Report-Spezifikation:</p> <p>Mit der Notation (<i>rep</i>) können Sie einen bestimmten anderen Report angeben, auf den sich das Statement beziehen soll.</p> <p>Es kann ein Wert von 0 bis 31 oder ein logischer Name, der mit einem DEFINE PRINTER-Statement zugewiesen wurde, angegeben werden. Falls (<i>rep</i>) nicht angegeben wird, bezieht sich das DISPLAY-Statement auf den ersten ausgegebenen Report (Report 0).</p> <p>Wenn diese Druckdatei für Natural als PC definiert ist, wird der Report auf den PC per Download heruntergeladen, siehe <i>Beispiel 8</i>.</p> <p>Weitere Informationen darüber, wie das Format eines Ausgabe-Reports gesteuert wird, siehe <i>Steuern der Ausgabe von Daten im Leitfaden zur Programmierung</i>.</p>
<i>options</i>	<p>Anzeige-Optionen:</p> <p>Einzelheiten siehe <i>Anzeige-Optionen</i> weiter unten.</p>
<i>output-format</i>	<p>Ausgabeformat-Definitionen:</p> <p>Einzelheiten siehe <i>Ausgabeformat-Definitionen</i> weiter unten.</p>
/	<p>Zeilenvorschub – Schrägstrich-Notation:</p> <p>Ein Schrägstrich (/) innerhalb eines Textelementes bewirkt einen Zeilenvorschub innerhalb des Textes.</p> <p>Ein Schrägstrich (/) zwischen zwei Ausgabeelementen bewirkt, dass das nachfolgende Element in derselben Spalte ausgegeben wird. Die Überschrift der Spalte wird gebildet, indem die Überschriften der beiden Elemente unmittelbar untereinander ausgegeben werden.</p> <p>Siehe auch die folgenden Themen im <i>Leitfaden zur Programmierung</i>:</p> <ul style="list-style-type: none"> ● <i>Zeilenvorschub – die Schrägstrich-Notation (/)</i> ● <i>Beispiel für Zeilenvorschub in DISPLAY-Statement</i> ● <i>Spaltenüberschriften unterdrücken – die Schrägstrich-Notation (‘/’)</i>
<i>output-element</i>	<p>Ausgabe-Element:</p> <p>Einzelheiten siehe <i>Ausgabe-Element</i> weiter unten.</p>

Anzeige-Optionen

<p>[NOTITLE] [NOHDR] [[AND] [GIVE] [SYSTEM] FUNCTIONS] [(statement-parameters)]</p>

Syntax-Element-Beschreibung:

NOTITLE	<p>Unterdrückung der Standard-Kopfzeile:</p> <p>Natural generiert für jede über ein DISPLAY-Statement ausgegebene Seite eine Standardkopfzeile. Diese Standardkopfzeile enthält die laufende Seitennummer, Datum und Uhrzeit. Die Uhrzeit wird zu Beginn der Programmausführung (TP-Betrieb) bzw. zu Beginn des Jobs (Batch-Betrieb) gesetzt. Die Standardkopfzeile kann mit einer eigenen Kopfzeile überschrieben werden, die mit einem WRITE TITLE-Statement angegeben wird. Das Schlüsselwort NOTITLE innerhalb des DISPLAY-Statements bewirkt, dass die Generierung einer Standardkopfzeile unterdrückt wird.</p> <p>Beispiele:</p> <ul style="list-style-type: none">● Generierte Standard-Titelzeile wird ausgegeben: <pre>DISPLAY NAME</pre>● Eigene Titelzeile wird ausgegeben: <pre>DISPLAY NAME WRITE TITLE 'user-title'</pre>● Keine Titelzeile wird ausgegeben: <pre>DISPLAY NOTITLE NAME</pre> <p>Anmerkung: Wird die NOTITLE-Option verwendet, gilt sie für alle DISPLAY-, PRINT- und WRITE-Statements im selben Objekt, die Daten auf denselben Report schreiben.</p>
----------------	----------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------

NOHDR	<p>Spaltenüberschriften:</p> <p>Für jede mittels eines DISPLAY-Statements ausgegebene Spalte von Feldwerten wird eine Spaltenüberschrift ausgegeben; hierbei gilt folgendes:</p> <ul style="list-style-type: none"> • Sie können mit dem DISPLAY-Statement explizit eine Spaltenüberschrift angeben, und zwar (in Apostrophen) vor dem jeweiligen Feldnamen. Zum Beispiel: <code>DISPLAY 'EMPLOYEE' NAME 'SALARY' SALARY</code> • Wenn Sie mit dem DISPLAY-Statement keine Spaltenüberschrift angeben, verwendet Natural die im DEFINE DATA-Statement für das Feld angegebene Spaltenüberschrift. • Wenn für ein Datenbankfeld im DEFINE DATA-Statement keine Spaltenüberschrift angegeben ist, wird die im betreffenden DDM definierte Standardüberschrift genommen. • Ist im DDM keine Spaltenüberschrift definiert, wird stattdessen der (im DDM definierte) Feldname als Überschrift verwendet. • Wenn für eine Benutzervariable im DEFINE DATA-Statement keine Spaltenüberschrift angegeben ist, wird der Variablenname als Überschrift verwendet. Zur Definition von Spaltenüberschriften vgl. DEFINE DATA-Statement. <code>DISPLAY NAME SALARY #NEW-SALARY</code> • Natural unterstreicht die Spaltenüberschriften immer und generiert zwischen Unterstreichung und den ausgegebenen Daten eine Leerzeile. • Enthält ein Programm mehrere DISPLAY-Statements, dann bestimmt das erste DISPLAY-Statement die Spaltenüberschriften; dies wird zur Kompilierungszeit ausgewertet. <p>Unterdrücken von Spaltenüberschriften:</p> <p>Um die Spaltenüberschrift für ein einzelnes Feld zu unterdrücken,</p> <ul style="list-style-type: none"> • geben Sie vor dem betreffenden Feldnamen die folgenden Zeichen (Apostroph-Schrägstrich-Apostroph) an: <code>'/'</code> <p>Beispiel: <code>DISPLAY '/' NAME 'SALARY' SALARY</code></p> <p>Sollen gar keine Spaltenüberschriften ausgegeben werden,</p> <ul style="list-style-type: none"> • geben Sie das Schlüsselwort NOHDR (für "No Header") an: <code>DISPLAY NOHDR NAME SALARY</code> <p>Anmerkungen:</p> <ol style="list-style-type: none"> 1. NOHDR gilt nur beim ersten DISPLAY-Statement, da nachfolgende DISPLAY-Statements keine Spaltenüberschriften erzeugen können. 2. Wenn Sie NOTITLE und NOHDR verwenden, müssen Sie sie in der folgenden Reihenfolge angeben: <code>DISPLAY NOTITLE NOHDR NAME SALARY</code>
-------	------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------

<p>GIVE SYSTEM FUNCTIONS</p>	<p>Benutzung von Systemfunktionen:</p> <p>Die GIVE SYSTEM FUNCTIONS-Klausel dient zur Auswertung der folgenden Natural-Systemfunktionen:</p> <p>AVER, COUNT, MAX, MIN, NAVER, NCOUNT, NMIN, SUM, TOTAL.</p> <p>Die Systemfunktionen werden ausgewertet, wenn das DISPLAY-Statement, das die GIVE SYSTEM FUNCTIONS-Klausel enthält, ausgeführt wird.</p> <p>Die Systemfunktionen können anschließend von einem Statement, das aufgrund einer End-of-Page-Bedingung ausgeführt wird, referenziert werden.</p> <p>Anmerkungen:</p> <ol style="list-style-type: none"> 1. Pro Report darf nur ein DISPLAY-Statement eine GIVE SYSTEM FUNCTIONS-Klausel enthalten. Die Auswertung von Systemfunktionen über DISPLAY GIVE SYSTEM FUNCTIONS geschieht seitenbezogen, d.h. bei Beginn einer neuen Seite werden alle Systemfunktionen außer TOTAL wieder auf Null gesetzt. 2. Bei der Verwendung von Systemfunktionen mit einem DISPLAY-Statement, das sich in einer Subroutine befindet, muss die End-of-Page-Verarbeitung innerhalb derselben Subroutine stattfinden. <p>Siehe auch <i>Beispiel 2 – DISPLAY-Statement mit GIVE SYSTEM FUNCTIONS-Klausel</i>.</p>
<p><i>statement-parameters</i></p>	<p>Parameter-Definition auf Statement-Ebene:</p> <p>Sie können (in Klammern) Session-Parameter auf Statement-Ebene setzen, die dann für das DISPLAY-Statement statt der betreffenden mit einem GLOBALS-Kommando, SET GLOBALS-Statement (nur im Reporting Mode) oder FORMAT-Statement gesetzten Parameter gelten.</p> <p>Werden mehrere Parameter angegeben, müssen sie jeweils durch ein oder mehrere Leerzeichen voneinander getrennt werden. Die Angabe eines Parameters darf sich nicht über zwei Sourcecode-Zeilen erstrecken.</p> <p>Anmerkung:</p> <p>Die hier gültigen Parameter-Einstellungen kommen nur für Variablenfelder in Betracht, haben aber keine Auswirkung auf Text-Konstanten. Möchten Sie Feldattribute für eine Text-Konstante setzen, müssen sie explizit für dieses Element gesetzt werden. Siehe <i>Parameter-Definition auf Elementebene (Feldebene)</i>.</p> <p>Siehe auch:</p> <ul style="list-style-type: none"> ● <i>Liste der Parameter</i> ● <i>Beispiel für die Benutzung von Parametern auf Statement- und Elementebene (Feldebene)</i> ● <i>Beispiel 7 – DISPLAY-Statement mit Parametern auf Statement-/Elementebene (Feldebene)</i>

Liste der Parameter

<p>Session-Parameter, die beim DISPLAY-Statement angegeben werden können</p>		<p>Spezifikation (S = auf Statement-Ebene, E = auf Elementebene)</p>
<p>AD</p>	<p>Attribute Definition</p>	<p>SE</p>
<p>AL</p>	<p>Alphanumeric Length for Output</p>	<p>SE</p>

Session-Parameter, die beim DISPLAY-Statement angegeben werden können		Spezifikation (S = auf Statement-Ebene, E = auf Elementebene)
BX	Box Definition	SE
CD	Color Definition	SE
CV	Control Variable	SE
DF	Date Format	SE
DL	Display Length for Output	SE
DY	Dynamic Attributes	SE
EM	Edit Mask	SE
ES	Empty Line Suppression	S
FC	Filler Character	SE
FL	Floating Point Mantissa Length	SE
GC	Filler Character for Group Headers	SE
HC	Header Centering	SE
HW	Heading Width	SE
IC	Insertion Character	SE
IS	Identical Suppress	SE
LC	Leading Characters	SE
LS	Line Size	S
MC	Multiple-Value Field Count	S
MP	Maximum Number of Pages of a Report	S
NL	Numeric Length for Output	SE
PC	Periodic Group Count	S
PM	Print Mode	SE
PS	Page Size	S
SF	Spacing Factor	SE
SG	Sign Position	SE
TC	Trailing Characters	SE
UC	Underlining Character	SE
ZP	Zero Printing	SE

Die einzelnen Parameter sind in der *Parameter-Referenz*-Dokumentation beschrieben.

Siehe auch die folgenden Themen im *Leitfaden zur Programmierung*:


- Spaltenüberschriften zentrieren – der HC-Parameter
- Breite der Spaltenüberschriften – der HW-Parameter
- Füllzeichen für Überschriften – die Parameter FC und GC
- Unterstreichungszeichen für Überschriften und Kopfzeilen – der UC-Parameter

Beispiel für die Benutzung von Parametern auf Statement- und Elementebene (Feldebene)

```

DEFINE DATA LOCAL
1 VARI (A4)      INIT <'1234'>
END-DEFINE
*
DISPLAY NOHDR          'Text'          '='    VARI          /*      Output
DISPLAY NOHDR (PM=I)  'Text'          '='    VARI          /*      Produced
*                                                              /*      -----
DISPLAY NOHDR          'Text'          '='    VARI          /*      Text 1234
DISPLAY NOHDR (PM=I)  'Text'          '='    VARI          /*      Text 4321
DISPLAY NOHDR          'Text' (PM=I)  '='    VARI (PM=I) /*      txeT 4321
DISPLAY NOHDR          'Text' (PM=I)  '='    VARI          /*      txeT 1234
END
    
```

Ausgabeformat-Definitionen



Feldpositionierung

<i>nX</i>	<p>Spaltenabstand:</p> <p>Mit dieser Notation fügen Sie zwischen den auszugebenden Spalten <i>n</i> Leerstellen ein. <i>n</i> darf nicht 0 sein.</p> <p>Beispiel:</p> <pre>DISPLAY NAME 5X SALARY</pre> <p>Siehe auch:</p> <ul style="list-style-type: none"> ● <i>Beispiel 1 – DISPLAY-Statement mit Notation nX- und nT</i> (weiter unten) ● <i>Spaltenabstand – der SF-Parameter und die Notation nX</i> (im Leitfaden zur Programmierung)
<i>nT</i>	<p>Setzen von Tabulatoren:</p> <p>Mit dieser Notation setzen Sie Tabulatoren, d.h. die Ausgabe eines Wertes beginnt ab Spalte <i>n</i>.</p> <p>Zurückpositionieren ist nicht erlaubt.</p> <p>Im folgenden Beispiel wird NAME ab Spalte 25 und SALARY ab Spalte 50 ausgegeben:</p> <pre>DISPLAY 25T NAME 50T SALARY</pre> <p>Siehe auch:</p> <ul style="list-style-type: none"> ● <i>Beispiel 1 – DISPLAY-Statement mit Notation nX- und nT</i> (weiter unten) ● <i>Tabulator-Notation nT</i> (im Leitfaden zur Programmierung)
<i>x/y</i>	<p><i>x/y</i>-Positionierung:</p> <p>Mit dieser Notation erreichen Sie, dass ein Feld <i>x</i> Zeilen unter der Ausgabe des letzten Statements, und zwar ab Spalte <i>y</i> ausgegeben wird.</p> <p><i>y</i> darf nicht 0 sein.</p> <p>Zurückpositionieren ist nicht erlaubt.</p>
<i>T*field-name</i>	<p>Feldbezogene Positionierung:</p> <p>Mit dieser Notation wird die Position eines Feldes nach der Position eines in einem vorhergehenden DISPLAY-Statement ausgegebenen Feldes (<i>field-name</i>) ausgerichtet.</p> <p>Zurückpositionieren ist nicht erlaubt.</p>


P*field-name	<p>Feld- und zeilenbezogene Positionierung:</p> <p>Mit dieser Notation werden Position und Ausgabezeile eines Feldes nach denen eines in einem vorhergehenden DISPLAY-Statement ausgegebenen Feldes (<i>field-name</i>) ausgerichtet. Dies wird meist bei vertikalen Ausgaben eingesetzt.</p> <p>Zurückpositionieren ist nicht erlaubt.</p> <p>Siehe auch:</p> <ul style="list-style-type: none"> ● <i>Beispiel 3 – DISPLAY-Statement mit der Notation P*</i> (weiter unten) ● <i>Tab-Notation P*field</i> (im Leitfaden zur Programmierung)
---------------------	-----------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------

Spaltenüberschriften

'text'	<p>Textzuweisung:</p> <p>In Apostrophen angegebener Text (<i>'text'</i>) vor einem Feld wird als Spaltenüberschrift verwendet. Ein Schrägstrich in Apostrophen <i>'/'</i> vor einem Feldnamen bewirkt, dass für dieses Feld keine Spaltenüberschrift ausgegeben wird. Beispiel:</p> <pre>DISPLAY 'EMPLOYEE' NAME 'MARITAL/STATUS' MAR-STAT</pre> <p>Werden vor einem Feldnamen mehrere Textelemente <i>'text'</i> angegeben, so wird das <i>letzte</i> als Spaltenüberschrift verwendet und die anderen werden in der Ausgabespalte vor dem Feldwert ausgegeben.</p> <p>Siehe auch:</p> <ul style="list-style-type: none"> ● <i>Eigene Spaltenüberschriften definieren</i> (im Leitfaden zur Programmierung) ● <i>Text Notation, Text-Notation, Text für ein Statement definieren</i> (im Leitfaden zur Programmierung) ● <i>Beispiel 4 – DISPLAY-Statement mit 'text', 'c'(n) und Attribut-Notation</i> (weiter unten)
'c'(n)	<p>Wiederholungszeichen:</p> <p>Das in Apostrophen (') stehende Zeichen <i>c</i> (<u>character</u>) wird <i>n</i>-mal unmittelbar vor dem Feldwert ausgegeben. Beispiel:</p> <pre>DISPLAY '*' (5) '=' NAME</pre> <p>führt zur Ausgabe von:</p> <pre>***** SMITH</pre> <p>Siehe auch:</p> <ul style="list-style-type: none"> ● <i>Text-Notation, n mal vor einem Feldwert anzuzeigendes Zeichen definieren</i> (im Leitfaden zur Programmierung) ● <i>Beispiel 4 – DISPLAY-Statement mit 'text', 'c(n)' und Attribut-Notation</i> (weiter unten)

Ausgabe-Attribute

attributes gibt die für die Text-Anzeige zu benutzenden Ausgabe-Attribute an. Es gibt die folgenden Attribute:



Die möglichen Parameterwerte sind in der *Parameter-Referenz* aufgeführt.

- *AD - Attribute Definition, Abschnitt Feldanzeige*
- *CD - Color Definition*
- *BX - Box Definition*
- *PM - Print Mode*

Anmerkung:

Der Compiler akzeptiert tatsächlich mehr als einen Attributwert für ein Ausgabefeld. Zum Beispiel können Sie Folgendes angeben: AD=BDI. In solch einem Fall gilt allerdings nur der letzte Wert. Im hier gezeigten Beispiel erhält nur der Wert I Gültigkeit, und das Ausgabefeld wird intensiviert (hell hervorgehoben) angezeigt.

Vertikale/Horizontale Ausgabe

Mit `DISPLAY VERT` werden die Werte mehrerer Felder nicht in Spalten nebeneinander sondern in einer Spalte untereinander ausgegeben. Eine neue Spalte wird durch Angabe des Schlüsselwortes `VERT` oder `HORIZ` initialisiert.

Die Ausgabe von Spaltenüberschriften wird beim `DISPLAY VERT` über die `AS`-Klausel gesteuert:

<u>VERTICALLY</u>	<p>Vertikale Spaltenausrichtung. Es wird keine Spaltenüberschrift erzeugt, wenn die AS-Klausel weggelassen wird.</p> <p>Beispiel:</p> <pre>DISPLAY VERT NAME SALARY</pre> <p>Siehe auch Beispiel <i>DISPLAY VERT ohne AS-Klausel</i> im <i>Leitfaden zur Programmierung</i>.</p>
AS 'text'	<p>Vertikale Spaltenausrichtung. Wenn AS 'text' angegeben wird, wird der in Apostrophen stehende Text als Spaltenüberschrift ausgegeben.</p> <p>Siehe auch Beispiel <i>DISPLAY VERT AS 'text'</i> im <i>Leitfaden zur Programmierung</i>.</p> <p>Wenn Sie einen Schrägstrich (/) in der Zeichenkette 'text' angeben, werden mehrere Zeilen mit Spaltenüberschriften erzeugt.</p> <p>Beispiel:</p> <pre>DISPLAY VERT AS 'LAST/NAME' NAME</pre>
AS 'text' CAPTIONED	<p>Vertikale Spaltenausrichtung. Wenn AS 'text' CAPTIONED angegeben wird, wird 'text' als Spaltenüberschrift ausgegeben, und außerdem wird die Standard-Spaltenüberschrift bzw. der Feldname in jeder Ausgabezeile dem jeweiligen Feldwert vorangestellt.</p> <p>Beispiel:</p> <pre>DISPLAY VERT AS 'PERSONS/SELECTED' CAPTIONED NAME FIRST-NAME</pre> <p>Siehe auch Beispiel <i>DISPLAY VERT AS 'text' CAPTIONED</i> im <i>Leitfaden zur Programmierung</i>.</p>
AS CAPTIONED	<p>Vertikale Spaltenausrichtung. Wenn AS CAPTIONED angegeben wird, wird der standardmäßige Überschriften-Text für das Feld ausgegeben (entweder Überschriften- Text oder den Feldnamen).</p> <p>Beispiel:</p> <pre>DISPLAY VERT AS CAPTIONED NAME FIRST-NAME</pre>
<u>HORIZONTALLY</u>	<p>Horizontale Spaltenausrichtung. Dies ist der standardmäßige Anzeigemodus.</p>

Vertikale und horizontale Ausgaben können miteinander kombiniert verwendet werden, wobei der Wechsel von einer Form zur anderen durch die Angabe des jeweiligen Schlüsselwortes (VERT oder HORIZ) erfolgt.

Um die vertikale Ausgabe für ein einzelnes Ausgabeelement auszusetzen, geben Sie vor dem Element einen Gedankenstrich (-) ein.

Beispiel:

DISPLAY VERT NAME - FIRST-NAME SALARY

würde bewirken, dass FIRST-NAME neben NAME ausgegeben wird, während SALARY wieder vertikal, d.h. unter NAME, ausgegeben wird.

Normalerweise erzeugt ein DISPLAY-Statement eine horizontale Ausgabe, d.h. die Ausgabe erfolgt in Spalten, die nebeneinander angeordnet sind.

Bei der Generierung der Spaltenüberschriften hat Natural folgende Prioritäten:

1. Der im DISPLAY-Statement für eine Spaltenüberschrift angegebene 'text'.
2. Bei Datenbankfeldern die im DDM definierte Standardspaltenüberschrift, bei Benutzervariablen der Variablenname.
3. Bei Datenbankfeldern der Name, unter dem das Feld im DDM definiert ist (wenn für das Datenbankfeld kein Überschriftentext definiert wurde).

Bei Gruppennamen wird eine Gruppen-Spaltenüberschrift für die gesamte Gruppe von Feldern erzeugt. Bei Angabe einer Gruppe kann nur diese Standard-Gruppenüberschrift durch eine eigene überschrieben werden.

Es sind bis zu 15 Spaltenüberschriftenzeilen erlaubt.

Die über ein DISPLAY-Statement erzeugte Ausgabe darf nicht über das Zeilenende hinausgehen; ist dies doch der Fall, gibt Natural eine entsprechende Fehlermeldung aus.

Weitere Informationen zur Benutzung der vertikalen/horizontalen Ausgabe siehe

- *Beispiel 5 – DISPLAY-Statement mit horizontaler Ausgabe*
- *Beispiel 6 – DISPLAY-Statement mit vertikaler und horizontaler Ausgabe*
- *DISPLAY VERT AS CAPTIONED und HORIZ im Leitfaden zur Programmierung*

Ausgabe-Element

<pre> [{ 'text' [(attributes)] } { 'c'(n) [(attributes)] } ... nX nT x/y </pre>	}	['='] {operand1 [(parameters)]}
------------------------------------------------------------------------------------------	---	---------------------------------

Operanden-Definitionstabelle:

Operand	Mögliche Struktur	Mögliche Formate	Referenzierung erlaubt	Dynam. Definition
<i>operand1</i>	S A G N	A N P I F B D T L G O	ja	nein

Syntax-Element-Beschreibung:

<i>nX</i>	Spalten-Abstand: Wie unter <i>Ausgabeformat-Definitionen</i> (siehe oben).
<i>nT</i>	Setzen von Tabulatoren: Wie unter <i>Ausgabeformat-Definitionen</i> (siehe oben).
<i>x/y</i>	<i>x/y</i>-Positionierung: Wie unter <i>Ausgabeformat-Definitionen</i> (siehe oben).
'<i>text</i>'	Textzuweisung: Wie unter <i>Ausgabeformat-Definitionen</i> (siehe oben).
'<i>c</i>'(<i>n</i>)	Wiederholungszeichen: Wie unter <i>Ausgabeformat-Definitionen</i> (siehe oben).
'<i>text</i>' '=' '<i>c</i>' (<i>n</i>) '='	Wird ' <i>text</i> ' '=' vor einem Feld angegeben, so wird <i>text</i> unmittelbar vor dem Feldwert ausgegeben. Dies gilt analog dazu für ' <i>c</i> ' (<i>n</i>) '='. Beispiel: DISPLAY '*****' '=' NAME
<i>attributes</i>	Ausgabe-Attribute: Wie unter <i>Ausgabeformat-Definitionen</i> (siehe oben).
<i>operand1</i>	Das auszugebende Feld. Anmerkung: Bei DL/I-Datenbanken: DL/I-AIX-Felder können nur angezeigt werden, wenn ein PCB verwendet wird, in dem das AIX im Parameter PROCSEQ angegeben ist. Ist dies nicht der Fall, gibt Natural zur Laufzeit eine Fehlermeldung aus.

<i>parameters</i>	<p>Parameter-Definiton auf Elementebene (Feldebene):</p> <p>Unmittelbar nach <i>operand1</i> können Sie auf Elementebene (Feldebene) in Klammern einen oder mehrere Parameter angeben, die dann für das betreffende Feld statt der entsprechenden auf Statement-Ebene oder mit einem GLOBALS-Kommando, SET GLOBALS-Statement (nur Reporting Mode)- oder FORMAT-Statement gesetzten Parameter gelten.</p> <p>Werden mehrere Parameter angegeben, müssen sie jeweils durch ein oder mehrere Leerzeichen voneinander getrennt werden. Die Angabe eines Parameters darf sich nicht über zwei Sourcecode-Zeilen erstrecken.</p> <p>Siehe auch:</p> <ul style="list-style-type: none"> ● <i>Liste der Parameter</i> ● <i>Beispiel für die Benutzung von Parametern auf Statement- und Elementebene (Feldebene)</i>
-------------------	---------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------

Standardwerte

Für ein DISPLAY-Statement gelten folgende Standardwerte:

- **Report-Breite:**

Die für Ausgaben gültige Standardbreite wird bei der Installation von Natural festgelegt; in der Regel beträgt sie im Batch-Betrieb 132 Stellen und entspricht im TP-Betrieb der Zeilenlänge des Terminals. Sie kann mit dem Session-Parameter LS überschrieben werden. Im TP-Betrieb setzt Natural die Parameter für Zeilenlänge (LS) und Seitenlänge (PS) unter Berücksichtigung der physischen Charakteristika des verwendeten Terminaltyps.

- **Terminal-BildschirmAusgabe:**

Erfolgt die DISPLAY-Ausgabe auf dem Bildschirm, dann beginnt die Ausgabe in der zweiten physischen Bildschirmspalte (da die erste Spalte für die etwaige Verwendung einer Attributstelle bei einem 3270-Terminal reserviert werden muss).

- **Druckausgabe auf Papier:**

Wird die DISPLAY-Ausgabe auf Papier ausgedruckt, dann beginnt die Ausgabe ganz links, d.h. in Spalte 1.

- **Abstandsfaktor:**

Standardmäßig wird zwischen zwei Ausgabeelementen eine Leerstelle eingefügt. Zwischen Ausgabespalten muss mindestens eine Leerspalte (reserviert für Terminal-Attribute) sein. Dieser Standardwert kann mit dem Session-Parameter SF überschrieben werden.

- **Feldausgabe:**

Die Breite einer Ausgabespalte richtet sich nach der Länge des Feldes oder der Spaltenüberschrift, je nachdem, was länger ist (es sei denn, der Parameter HW wird verwendet).

- Ist die Überschrift kürzer als das Feld, wird sie über der Spalte zentriert (es sei denn, mit dem Parameter HC=L bzw. HC=R wird eine linksbündige bzw. rechtsbündige Ausgabe veranlasst).

- Ist das Feld kürzer als die Überschrift, wird das Feld linksbündig zur Überschrift ausgerichtet.
 - Bei alphanumerischen Feldern werden die Feldwerte linksbündig im Feld ausgegeben, bei numerischen rechtsbündig.
 - Mit dem Parameter AD=L kann auch bei numerischen Feldern eine linksbündige Ausgabe erreicht werden.
 - Mit dem Parameter AD=R kann bei alphanumerischen Feldern eine rechtsbündige Ausgabe erreicht werden.
 - Bei vertikalen Ausgaben richtet sich die Breite einer Spalte nach dem längsten Feldwert bzw. der längsten Überschrift (es sei denn, der Parameter HW wird verwendet).
- **Vorzeichen:**
Bei der Ausgabe eines numerischen Feldes wird eine Stelle vor dem Feld für die Ausgabe eines Vorzeichens reserviert. Die Vorzeichenstelle kann mit dem Session-Parameter SG unterdrückt werden.
 - **Seitenumbruch:**
Natural prüft vor der Ausführung eines DISPLAY-Statements, wann ein Seitenumbruch erforderlich ist. *Während* der Ausführung des DISPLAY-Statements werden keine Kopf- oder Fußzeilen generiert.

Beispiele

- Beispiel 1 — DISPLAY-Statement mit der Notation nX und nT
- Beispiel 2 — DISPLAY-Statement mit GIVE SYSTEM FUNCTIONS-Klausel
- Beispiel 3 — DISPLAY-Statement mit der Notation P*
- Beispiel 4 — DISPLAY-Statement mit 'text', 'c(n)' und Attribut-Notation
- Beispiel 5 — DISPLAY-Statement mit horizontaler Ausgabe
- Beispiel 6 — DISPLAY-Statement mit vertikaler und horizontaler Ausgabe
- Beispiel 7 — DISPLAY-Statement mit Parametern auf Statement-/Elementebene (Feldebene)
- Beispiel 8 — Report-Spezifikation mit für Natural als PC definierter Ausgabedatei

Beispiel 1 — DISPLAY-Statement mit der Notation nX und nT

```

** Example 'DISEX1': DISPLAY (with nX, nT notation)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 JOB-TITLE
END-DEFINE
*
LIMIT 4
READ EMPL-VIEW BY NAME

```

```

    DISPLAY NOTITLE 5X NAME 50T JOB-TITLE
END-READ
*
END

```

Ausgabe des Programms DISEX1:

NAME	CURRENT POSITION
ABELLAN	MAQUINISTA
ACHIESON	DATA BASE ADMINISTRATOR
ADAM	CHEF DE SERVICE
ADKINSON	PROGRAMMER

Beispiel 2 — DISPLAY-Statement mit GIVE SYSTEM FUNCTIONS-Klausel

```

** Example 'DISEX2': DISPLAY (with GIVE SYSTEM FUNCTIONS)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 NAME
  2 FIRST-NAME
  2 SALARY      (1)
  2 CURR-CODE  (1)
END-DEFINE
*
LIMIT 15
FORMAT PS=15
*
READ EMPLOY-VIEW
  DISPLAY GIVE SYSTEM FUNCTIONS
    PERSONNEL-ID NAME FIRST-NAME SALARY (1) CURR-CODE (1)
  AT END OF PAGE
    WRITE / 'SALARY STATISTICS:'
      / 7X 'MAXIMUM:' MAX(SALARY(1)) CURR-CODE (1)
      / 7X 'MINIMUM:' MIN(SALARY(1)) CURR-CODE (1)
      / 7X 'AVERAGE:' AVER(SALARY(1)) CURR-CODE (1)
  END-ENDPAGE
END-READ
*
END

```

Ausgabe des Programms DISEX2:

Page 1 05-01-12 09:47:48

PERSONNEL ID	NAME	FIRST-NAME	ANNUAL SALARY	CURRENCY CODE
50005500	BLOND	ALEXANDRE	172000	FRA
50005300	MAIZIERE	ELISABETH	166900	FRA
50004900	CAUDAL	ALBERT	167350	FRA
50004600	VERDIE	BERNARD	170100	FRA
50004200	VAUZELLE	BERNARD	159790	FRA
50004100	CHAPUIS	ROBERT	169900	FRA
50003800	JOUSSELIN	DANIEL	171990	FRA
50006900	BAILLET	PATRICK	188000	FRA

```
50007600  MARX                JEAN-MARIE                365700  FRA

SALARY STATISTICS:
  MAXIMUM:    365700  FRA
  MINIMUM:    159790  FRA
  AVERAGE:   192414  FRA
```

Beispiel 3 — DISPLAY-Statement mit der Notation P*

```
** Example 'DISEX3': DISPLAY (with P* notation)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 SALARY (1)
  2 BIRTH
  2 CITY
END-DEFINE
*
LIMIT 2
READ EMPL-VIEW BY CITY FROM 'N'
  DISPLAY NOTITLE NAME CITY
    VERT AS 'BIRTH/SALARY' BIRTH (EM=YYYY-MM-DD) SALARY (1)
  SKIP 1
  AT BREAK OF CITY
    DISPLAY P*SALARY (1) AVER(SALARY (1))
  SKIP 1
  END-BREAK
END-READ
END
```

Ausgabe des Programms DISEX3:

NAME	CITY	BIRTH	SALARY
WILCOX	NASHVILLE	1970-01-01	38000
MORRISON	NASHVILLE	1949-07-10	36000
			37000

Beispiel 4 — DISPLAY-Statement mit 'text', 'c(n)' und Attribut-Notation

```
** Example 'DISEX4': DISPLAY (with 'c(n)' notation and attribute)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 DEPT
  2 LEAVE-DUE
  2 NAME
END-DEFINE
*
LIMIT 4
READ EMPL-VIEW BY DEPT FROM 'T'
  IF LEAVE-DUE GT 40
    DISPLAY NOTITLE
```

```

        'EMPLOYEE' NAME           /* OVERRIDE STANDARD HEADER
        'LEAVE ACCUMULATED' LEAVE-DUE /* OVERRIDE STANDARD HEADER
        '**(10)(I)                /* DISPLAY 10 '** INTENSIFIED
ELSE
    DISPLAY NAME LEAVE-DUE
END-IF
END-READ
*
END

```

Ausgabe des Programms DISEX4:

EMPLOYEE	LEAVE ACCUMULATED
LAVENDA	33
BOYER	33
CORREARD	45
BOUVIER	19

Beispiel 5 — DISPLAY-Statement mit horizontaler Ausgabe

```

** Example 'DISEX5': DISPLAY (horizontal display)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 JOB-TITLE
  2 SALARY (1:2)
  2 CURR-CODE (1:2)
END-DEFINE
*
LIMIT 4
READ EMPL-VIEW BY NAME
  DISPLAY NOTITLE NAME JOB-TITLE SALARY (1:2) CURR-CODE (1:2)
  SKIP 1
END-READ
*
END

```

Ausgabe des Programms DISEX5:

NAME	CURRENT POSITION	ANNUAL SALARY	CURRENCY CODE
ABELLAN	MAQUINISTA	1450000	PTA
		1392000	PTA
ACHIESON	DATA BASE ADMINISTRATOR	11300	UKL
		10500	UKL
ADAM	CHEF DE SERVICE	159980	FRA
		0	
ADKINSON	PROGRAMMER	34500	USD
		31700	USD

Beispiel 6 — DISPLAY-Statement mit vertikaler und horizontaler Ausgabe

```

** Example 'DISEX6': DISPLAY (vertical and horizontal display)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 CITY
  2 JOB-TITLE
  2 SALARY (1:2)
  2 CURR-CODE (1:2)
END-DEFINE
*
LIMIT 1
READ EMPL-VIEW BY NAME
  DISPLAY NOTITLE VERT AS CAPTIONED
    NAME CITY 'POSITION' JOB-TITLE
    HORIZ 'SALARY' SALARY (1:2) 'CURRENCY' CURR-CODE (1:2)
  /*
  SKIP 1
END-READ
END

```

Ausgabe des Programms DISEX6:

NAME CITY POSITION	SALARY	CURRENCY
ABELLAN	1450000	PTA
MADRID	1392000	PTA
MAQUINISTA		

Beispiel 7 — DISPLAY-Statement mit Parametern auf Statement-/Elementebene (Feldebene)

```

** Example 'DISEX7': DISPLAY (with parameters for statement/element)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 PERSONNEL-ID
  2 TELEPHONE
  3 AREA-CODE
  3 PHONE
END-DEFINE
*
LIMIT 3
READ EMPL-VIEW BY NAME
  DISPLAY NOTITLE (AL=16 GC=+ NL=8 SF=3 UC==)
    PERSONNEL-ID NAME TELEPHONE (LC=< TC=>)
END-READ
END

```

Ausgabe des Programms DISEX7:

PERSONNEL ID	NAME	AREA CODE	TELEPHONE
60008339	ABELLAN	<1 >	<4356726 >
30000231	ACHIESON	<0332 >	<523341 >
50005800	ADAM	<1033 >	<44864858 >

Beispiel 8 — Report-Spezifikation mit für Natural als PC definierter Ausgabedatei

```

** Example 'PCDIEX1': DISPLAY and WRITE to PC
**
** NOTE: Example requires that Natural Connection is installed.
*****
DEFINE DATA LOCAL
01 PERS VIEW OF EMPLOYEES
  02 PERSONNEL-ID
  02 NAME
  02 CITY
END-DEFINE
*
FIND PERS WITH CITY = 'NEW YORK' /* Data selection
WRITE (7) TITLE LEFT 'List of employees in New York' /
  DISPLAY (7) /* (7) designates the output file (here the PC).
  'Location' CITY
  'Surname' NAME
  'ID' PERSONNEL-ID
END-FIND
END

```